

Rust コンパイラーウォークスルー

原 将己

2017 年 5 月 5 日

目次

第 1 章	はじめに	5
第 2 章	Rust コンパイラーの開発フロー	7
2.1	RFC	7
2.2	メインリポジトリ	7
2.3	文書	7
2.4	フォーラム	7
第 3 章	Rustc と Crate と Cargo	9
3.1	crate のソース	9
3.2	crate のバイナリー	9
3.3	Cargo の基本動作	9
3.4	依存関係の解決	9
3.5	crates リポジトリ	9
第 4 章	字句と構文	11
4.1	字句解析	11
4.2	構文解析	21
第 5 章	構文拡張	31
5.1	構文拡張のデータベース	31
5.2	マクロ	31
5.3	組み込みマクロ	31
5.4	macro_rules	31
5.5	属性構文拡張	31
5.6	手続きマクロ	31
第 6 章	名前解決	33
6.1	モジュール	33
6.2	インポート解決	33
6.3	レキシカルスコープの解決	33
6.4	型依存の名前解決	33
6.5	生存期間の解決	33
第 7 章	型検査	35

7.1	型	35
7.2	トレイト	35
7.3	Hindley-Milner 型推論	35
7.4	トレイト選択	35
7.5	リージョン推論	35
7.6	ドロップ検査	35
7.7	借用検査	35
7.8	可変性検査	35
第 8 章	コード生成	37
8.1	MIR	37
8.2	LLVM IR	37
第 9 章	ドキュメンテーション	39
9.1	doc-comment の構文	39
参考文献		41

第 1 章

はじめに

第 2 章

Rust コンパイラーの開発フロー

2.1 RFC

2.2 メインリポジトリ

2.3 文書

2.4 フォーラム

第 3 章

Rustc と Crate と Cargo

3.1 crate のソース

3.2 crate のバイナリー

3.3 Cargo の基本動作

3.4 依存関係の解決

3.5 crates リポジトリ

第 4 章

字句と構文

4.1 字句解析

Rust のソースファイルはまず字句解析にかけられます。

4.1.1 凡例

この節では BNF 風の記法を用いて文法を記述しています。BNF 風ですが実際には PEG (解析表現文法) に近い意味論です。

- **A B** は接続です。2 つの構文要素をこの順に並べたものです。
- **A | B** は PEG の順序つき OR (**A / B**) です。**A** を試し、失敗したら **B** を試します。ただし、この節で **A | B** と書いた場合は、順番による違いが生じないようになっているはずです。
- **A***, **A+**, **A?** は、0 回以上の繰り返し、1 回以上の繰り返し、高々 1 回の出現です。いずれも貪欲に実行したものと解釈しますが、それを知らなくても読めるように配慮したつもりです。
- **lookahead A** と **not A** は先読み肯定と先読み否定です。慣れていないと読み間違えやすいかもしれません。**A (not B)** は、**A** だが **B** が後続しないものです。**(not B) A** は、**A** だが同じ場所で **B** にはならないものです。
- **eof** は文字列の末尾です。
- **"+="** のようにダブルクォートで囲んだものは文字列で、1 文字ずつ接続したものと同じです。C 言語風のエスケープを使っています。

4.1.2 Unicode

Rust の字句解析器は UTF-8 でエンコードされたファイルを入力します。UTF-8 をデコードしたものは、Unicode スカラー値の列とみなすことができます。

Unicode スカラー値とは、0 以上 0xD800 未満か 0xE000 以上 0x10FFFF 未満の整数のことです。ここでは、Unicode 文字といったら Unicode スカラー値のことを指すことにします。

```
1 AnyChar ::= /* Any Unicode scalar value */
2
```

```

3 AnyAsciiChar ::=
4   /*Any Unicode scalar value < 128 */

```

4.1.3 トークン列

Rust の字句解析器の仕事は、Unicode スカラー値の列を受け取り、字句エラーがなければトークン列を返すことです。

```

1 File ::= MaybeShebang Skip (Token Skip)* eof
2
3 Token ::=
4     InnerDocComment
5   | OuterDocComment
6   | Symbol
7   | Literal
8   | Identifier
9   | StrictKeyword
10  | ReservedKeyword
11  | Lifetime

```

- 字句的な特徴からトークンを分類すると、doc-comment, 記号、数値リテラル、文字列リテラル、識別子とキーワード、生存期間と分けられます。しかし、構文解析器やマクロ展開器にとっての分類は必ずしもこれに沿った形ではありません。

4.1.4 空白とコメント

Rust の字句解析器は空白とコメントを無視します。コンパイラにとっては、これらはトークンを明示的に分割する以上の役割はありません。ただし、doc-comment は特別な扱いを受けます。

```

1 Skip ::= (Whitespace | Comment)* (not MaySkip)
2
3 Whitespace ::= PatternWhitespace+ (not PatternWhitespace)
4
5 PatternWhitespace ::=
6   /* Any Unicode scalar value with PATTERN_WHITE_SPACE */
7
8 Comment ::= "//" (not ("/" | "!")) LineCommentChar* ("\n" | eof)
9           | "///" LineCommentChar* ("\n" | eof)
10          | "/*" (not ("*" | "!")) BlockCommentBody* "*/"
11          | "/**/"
12
13 LineCommentChar ::= (not "\n") AnyChar
14 BlockCommentBody ::= (not ("/*" | "*/")) AnyChar
15                   | "/*" BlockCommentBody* "*/"
16
17 MaySkip ::= PatternWhitespace

```

```

18 | "/" (not ("/" | "!"))
19 | "////"
20 | "/*" (not ("*" | "!"))
21 | "*/"
22
23 MaybeShebang ::= Shebang | not Shebang
24
25 Shebang ::= "#!" (not "[") LineCommentChar* ("\n" | eof)

```

- UAX #31 [6] に従い、`Pattern_White_Space` に分類される文字は全て空白文字とみなされます。`Pattern_White_Space` は UAX #44 [7] で規定されており、`PropList.txt` で取得できます。現時点では以下の文字が `Pattern_White_Space` に分類されます。

- U+0009 Control: CHARACTER TABULATION
- U+000A Control: LINE FEED (LF)
- U+000B Control: LINE TABULATION
- U+000C Control: FORM FEED (FF)
- U+000D Control: CARRIAGE RETURN (CR)
- U+0020 SPACE
- U+0085 Control: NEXT LINE (NEL)
- U+200E LEFT-TO-RIGHT MARK
- U+200F RIGHT-TO-LEFT MARK
- U+2028 LINE SEPARATOR
- U+2029 PARAGRAPH SEPARATOR

- C と同様に `//` と `/*` がコメントとみなされますが、`doc-comment` はトークンとみなされるため、この意味で通常のコメントとは異なります。
- C とは異なり、`/*` はネストさせることができます。
- `////` 型のコメントの中では、`doc-comment` と同様に、単独でのキャリッジリターンの出現が禁止されています。これはバグではないかと思います。
- 1 行目が `#!` で始まる場合、その行はシバンとみなされ、コメントと同様に無視されます。ただし、`#!` で始まる場合は、内部属性と紛らわしいため、例外的にシバンとはみなされません。

4.1.5 doc-comment

`doc-comment` はコメントとよく似た構文を持ちますが、Rust コンパイラはこれを一つのトークンとみなします。

```

1 InnerDocComment ::=
2   "//!" LineDocCommentChar* ("\r\n" | "\n" | eof)
3   | "/*!" BlockDocCommentBody* "*/"
4
5 OuterDocComment ::=
6   "///" (not "/") LineDocCommentChar* ("\r\n" | "\n" | eof)

```

```

7 | "/*" (not "/" ) BlockDocCommentBody* "*/"
8
9 LineDocCommentChar ::= (not ("\r" | "\n")) AnyChar
10
11
12 BlockDocCommentBody ::= (not ("\r" | "/*" | "*/")) AnyChar
13 | "\r\n"
14 | "/*" BlockDocCommentBody* "*/"

```

- ///, //!, /*, /*! で始まるコメントは、doc-comment と解釈されます。
- 例外として、//// で始まるコメントと /**/ は、doc-comment ではありません。
- これらの doc-comment は、外部ツールだけではなく、Rust コンパイラによって認識されます。字句的に正しい doc-comment であっても、構文的に誤った位置にあれば、コンパイルエラーになる可能性があります。
- 通常のコメントと異なり、doc-comment 内ではキャリッジリターン (CR) が単独で出現してはいけません。必ずラインフィードと対で (CRLF) 出現する必要があります。
- C とは異なり、/* はネストさせることができます。

4.1.6 記号

```

1 Symbol ::=
2   "=" (not ("=" | ">"))
3   | "<" (not ("=" | "<" | "-"))
4   | "<=" | "==" | "!=" | ">="
5   | ">" (not ("=" | ">"))
6   | "&&" | "||"
7   | "!" (not "=")
8   | "~"
9   | "+" (not "=")
10  | "-" (not ("=" | ">"))
11  | "*" (not "=")
12  | "/" (not "=")
13  | "^" (not "=")
14  | "&" (not ("=" | "&"))
15  | "|" (not ("=" | "|"))
16  | ">>" (not "=")
17  | "<<" (not "=")
18  | "+=" | "-=" | "*=" | "/=" | "^="
19  | "&=" | "|=" | ">>=" | "<<="
20  | "@"
21  | "." (not ".")
22  | ".." (not ".")
23  | "... " | "," | ";"
24  | ":" (not ":")
25  | "::" | "->" | "<- " | "=>" | "#"

```

```

26 | "$" | "?"
27 | "(" | ")" | "{" | "}" | "[" | "]"
28 | Underscore

```

- 記号のパーズ規則は単純です。記号の候補の中で、一番長いものを選択します。
- アンダースコアだけ特殊です。アンダースコアは字句的には識別子に近いですが、パーサーに渡す段階では識別子よりも記号に近い扱いを受けます。

4.1.7 リテラル

```

1 Literal ::= LiteralBody LiteralSuffix?
2
3 LiteralBody ::= NumberLiteralBody | StringLikeLiteralBody
4
5 LiteralSuffix ::= IdentifierOrKeyword | Underscore

```

- リテラルは、数値リテラルと、文字列系リテラルの2種類あります。
- いずれのリテラルにも、識別子またはキーワードを後置することができます。これは型を表すのに使います。
- 識別子またはキーワードのかわりにアンダースコアを後置することもできます。リテラルの構造上数値リテラルでは不可能です。後置しても、パーサーからはなかったものとして扱われます。もしかしたらバグかもしれません。
- 後置された文字列は型として解釈されますが、これは構文解析器の仕事です。

4.1.8 数値リテラル

```

1 NumberLiteralBody ::= IntegerLiteralBody | FloatLiteralBody
2
3 IntegerLiteralBody ::=
4   "0b" "_"* BinDigit BinDigitU* (not DecDigitU) EnsureNotFloat
5 | "0o" "_"* OctDigit OctDigitU* (not DecDigitU) EnsureNotFloat
6 | Decimal (not DecDigitU) EnsureNotFloat
7 | "0x" "_"* HexDigit HexDigitU* (not HexDigitU) EnsureNotFloat
8
9 EnsureNotFloat ::= not ( "." | "e" | "E" )
10                  | lookahead " ."
11                  | lookahead ( "." IdentStart )
12
13 FloatLiteralBody ::=
14   Decimal " ." (not ( "." | IdentStart | DecDigit ))
15 | Decimal " ." Decimal (not ( "e" | "E" ))
16 | Decimal " ." Decimal ExponentPart
17 | Decimal ExponentPart
18
19 ExponentPart ::=
20   ( "e" | "E" ) ( "-" | "+" )?

```

```

21     "_"* DecDigit DecDigitU* (not DecDigitU)
22
23 Decimal ::= DecDigit DecDigitU*
24
25 BinDigit ::= "0" | "1"
26
27 OctDigit ::= "0" | "1" | "2" | "3" | "4" | "5" | "6" | "7"
28
29 DecDigit ::= OctDigit | "8" | "9"
30
31 HexDigit ::=
32     DecDigit | "A" | "B" | "C" | "D" | "E" | "F"
33     | "a" | "b" | "c" | "d" | "e" | "f"
34
35 BinDigitU ::= BinDigit | "_"
36
37 OctDigitU ::= OctDigit | "_"
38
39 DecDigitU ::= DecDigit | "_"
40
41 HexDigitU ::= HexDigit | "_"

```

- 数値リテラルは、整数リテラルと浮動小数点数リテラルからなります。
- 負の符号は単なる単項演算子です。トークンとしては、非負の数のみ扱われます。
- アンダースコアで数を区切ることができます。しかし、数字がないのにアンダースコアだけある、という状況は禁止されています。また、リテラルの先頭や、小数点の直後には、アンダースコアを置くことはできません。
- 2進数、8進数、16進数では接頭辞を明示する必要があります。0で始まっていても10進数です。また、接頭辞は小文字でなければなりません。16進数の桁や、指数部をあらわすeは大文字でも小文字でも構いません。
- 小数点の手前を省略することはできません。また、小数点の直後を省略できるのは、直後が.でも識別子でもない場合だけです。
- 整数の直後に.を置けるのは、その直後が.または識別子の場合だけです。
- 上記の規則について、現在は._が特別扱いされています。おそらくバグではないかと思います。
- また、現在は0e, 0Eから始まる浮動小数点数がパースエラーになるバグがあります。これは修正済みで、次の安定板には入ると思います。
- aからfまでの文字は、16進数以外では、トークン区切りとみなされます。しかし、たとえ8進数であっても、9はトークン区切りとはみなされません。この位置でパースエラーになります。
- fという文字は、微妙なバランスの上に成立しています。16進数は整数でしか使えないので、接尾辞にはiかuが指定できれば十分です。いっぽう、fを含む接尾辞が指定される浮動小数点数では、fの部分でリテラル本体のパースが打ち切られるため、うまくfを指定することができます。

4.1.9 文字列系リテラル

```

1 StringLikeLiteralBody ::=
2     ByteLiteralBody | ByteStringLiteralBody
3     | CharLiteralBody | CharStringLiteralBody
4
5 ByteLiteralBody ::=
6     "b\'" (not ("\" | "\t" | "\n" | "\r" | "\'")) AnyAsciiChar "\'"
7     | "b\'\" ("n" | "r" | "t" | "\" | "\'" | "\"" | "0") "\'"
8     | "b\'\"x" HexDigit HexDigit "\'"
9
10 ByteStringLiteralBody ::=
11     "b\" SingleByte* "\""
12     | "br" repeated("#", N) "\""
13     ((not ("\" repeated("#", N))) AnyAsciiChar)*
14     "\" repeated("#", N)
15     /* N >= 0 */
16     /* repeated("#", 3) means e.g. "###" */
17
18 SingleByte ::=
19     (not ("\" | "\r" | "\")) AnyAsciiChar
20     | "\r\n"
21     | "\" ("n" | "r" | "t" | "\" | "\'" | "\"" | "0")
22     | "\" ("\r\n" | "\n") Whitespace
23     | "\"x" HexDigit HexDigit
24
25 CharLiteralBody ::=
26     "\' (not ("\" | "\t" | "\n" | "\r" | "\'")) AnyChar "\'"
27     | "\'\" ("n" | "r" | "t" | "\" | "\'" | "\"" | "0") "\'"
28     | "\'\"x" OctDigit HexDigit "\'"
29     | "\'\"u{" HexDigit{1, 6} "}" "\'"
30     /* only when x < 0xD800 or 0xE000 < x < 0x110000 */
31
32 CharStringLiteralBody ::=
33     "\" SingleChar* "\""
34     | "r" repeated("#", N) "\""
35     ((not ("\" repeated("#", N))) AnyChar)*
36     "\" repeated("#", N)
37     /* N >= 0 */
38     /* repeated("#", 3) means e.g. "###" */
39
40 SingleChar ::=
41     (not ("\" | "\r" | "\")) AnyChar
42     | "\r\n"
43     | "\" ("n" | "r" | "t" | "\" | "\'" | "\"" | "0")
44     | "\" ("\r\n" | "\n") Whitespace

```

```

45 | "\\x" OctDigit HexDigit
46 | "\\u{" HexDigit{1, 6} "}"
47 /* only when x < 0xD800 or 0xE000 < x < 0x110000 */

```

- バイト、バイト列、文字、文字列のいずれかを表します。
- バイトまたは文字は ' で囲み、バイト列または文字列は " で囲みます。
- バイトまたはバイト列をあらわすときは **b** を前置します。
- バイト列または文字列の場合は生リテラルを使うこともできます。生リテラルは " のかわりに **r###** のような形で開始します。この場合最初に **###** が出現した位置で終了になります。# の個数は 0 個以上で、左右で揃える必要があります。
- バックスラッシュ、水平タブ、ラインフィード、キャリッジリターン、デリミタ (" または ' のうちリテラルを囲うのに使われているほう) 以外の ASCII 文字は、そのまま書くことができます。
- ASCII 以外の Unicode 文字 (Unicode スカラー値) は、文字または文字列リテラルにそのまま書くことができます。
- タブ、ラインフィード、CRLF (キャリッジリターン + ラインフィード) は、生リテラルでないバイト列または文字列に直接書くことができます。CRLF は LF とみなされます。
- 生リテラル以外のリテラルでは、**\n**, **\r**, **\t**, ****, **\'**, **\"**, **\0** がエスケープとして利用できます。
- 生リテラル以外のリテラルでは、**\x** に続けて 16 進法で 2 桁 (アンダースコア不可) の整数を指定することで、1 バイトエスケープになります。ただし、128 より大きいバイトは、文字または文字列リテラルでは指定できません。
- 生リテラル以外の文字または文字列リテラルでは、**\u{}** の中に 16 進法で 1 桁から 6 桁まで (アンダースコア不可) の整数を指定することで、Unicode エスケープになります。ただし指定できるのは Unicode スカラー値、つまり 0 以上 0xD800 未満か、0xE000 以上 0x110000 未満の整数のみです。
- 生リテラル以外のバイト列または文字列リテラルでは、バックスラッシュの直後に改行 (LF または CRLF) を続けると、該当バックスラッシュから、連続する空白文字が全てスキップされます。
- 生リテラルではエスケープは使用できず、全ての文字がそのまま解釈されます。生バイト列リテラルでは、ASCII 文字以外 (128 以上のスカラー値) は使用できません。

4.1.10 識別子またはキーワード

```

1 Identifier ::=
2 (not (StrictKeyword | ReservedKeyword)) IdentifierOrKeyword
3
4 IdentifierOrKeyword ::=
5 (not (Underscore | MaybeString))
6 IdentStart IdentContinue* (not IdentContinue)
7

```

```

8 IdentStart ::=
9   /* Any Unicode scalar value with XID_Start */ | "_"
10
11 IdentContinue ::=
12   /* Any Unicode scalar value with XID_Continue */ | "_"
13
14 Underscore ::= "_" (not IdentContinue)
15
16 MaybeString ::= "r\"" | "r#" | "b\"" | "b\'" | "br\"" | "br#"
17
18 StrictKeyword ::= StrictKeywordString (not IdentContinue)
19
20 StrictKeywordString ::=
21   "as" | "box" | "break" | "const" | "continue" | "crate"
22   | "else" | "enum" | "extern" | "false" | "fn" | "for"
23   | "if" | "impl" | "in" | "let" | "loop" | "match" | "mod"
24   | "move" | "mut" | "pub" | "ref" | "return" | "self" | "Self"
25   | "static" | "struct" | "super" | "trait" | "true" | "type"
26   | "unsafe" | "use" | "where" | "while"
27
28 ReservedKeyword ::= ReservedKeywordString (not IdentContinue)
29
30 ReservedKeywordString ::=
31   "abstract" | "alignof" | "become" | "do" | "final"
32   | "macro" | "offsetof" | "override" | "priv" | "proc"
33   | "pure" | "sizeof" | "typeof" | "unsized" | "virtual"
34   | "yield"
35
36 WeakKeyword ::= WeakKeywordString (not IdentContinue)
37
38 WeakKeywordString ::= "default" | "union"

```

- 識別子またはキーワードは、ASCII の数字、大文字、小文字、アンダースコアからなります。また最初の文字は ASCII の大文字、小文字、アンダースコアのいずれかでなければなりません。
- この規則は `feature(non_ascii_idents)` [1] により Unicode に拡張されます。この場合、識別子は `XID_Continue` な文字またはアンダースコアからなります。また、最初の文字は `XID_Start` な文字またはアンダースコアでなければなりません。これは UAX #31 [6] を考慮していますが、現在の実装は完全とはいえません。UAX #31 では NKFC 等の正規化が言及されていますが、現在のコンパイラは正規化を全くしません。どの正規化をするかなどを含めて、議論の途中の状態です。
- アンダースコア 1 文字からなる場合は、識別子ではなく記号として解釈されます。
- `r`", `r#`", `b`", `b'`", `br`", `br#` のいずれかで始まる場合は、これらの `r`, `b`, `br` は識別子ではなく文字列リテラルの開始として解釈されます。
- 「識別子またはキーワード」のうち、いくつかは「強キーワード (strict keyword)」 「予約キーワード (reserved keyword)」 「弱キーワード (weak keyword)」 のいずれか

に分類されています。強キーワードは特別な意味をもつもので、予約キーワードは将来強キーワードとして使われる可能性を考慮して予約されているものです。

- 強キーワードでも予約キーワードでもないものを識別子といいます。弱キーワードは、原則として識別子として扱われますが、文脈によっては特別な意味を持ちます。
- 識別子またはキーワードは、マクロの観点からはキーワードか否かに関係なく同等に扱われます。識別子またはキーワードと、生存期間には、構文文脈が付与されます。アンダースコアは識別子またはキーワードではないため、構文文脈は付与されません。
- 字句解析器は、空の構文文脈を付与します。

4.1.11 生存期間

```

1 Lifetime ::=
2   (not "\" IdentStart "\"")
3   "\" IdentifierOrKeyword
4
5 WeakLifetimeKeyword ::= "\"static" (not IdentContinue)
```

- ' に識別子またはキーワードを後続させたものは、生存期間です。
- ただし、文字リテラルとしても解釈できる場合は、文字リテラルが優先されます。
- '_ は生存期間ではありません。
- 'static は生存期間ですが、文脈によっては特別な意味を持ちます。
- 識別子またはキーワードと、生存期間には、構文文脈が付与されます。
- 字句解析器は、空の構文文脈を付与します。

4.1.12 特殊なトークン

字句解析器以外によってのみ生成される特殊なトークンがいくつかあります。

- 空文字列、`{{root}}`、`$crate` はいずれも、上で説明した「識別子またはキーワード」の条件を満たしていませんが、内部的に識別子として生成される場合があります。
- 同じく、空文字列が内部的に生存期間として生成される場合があります。なお、通常の生存期間の名前は、内部的には ' を含んだ形で保存されています。
- 補間トークンは、AST を 1 つのトークンとしてみなしたものです。
 - アイテム、ブロック、文、パターン、式、型、識別子を含む補間トークンは、マクロの展開のために使われます。
 - メタアイテム、パス、トークンツリーを含む補間トークンは、属性構文拡張の展開のために使われます。
 - パターンマッチの腕、実装内アイテム、トレイト内アイテム、ジェネリックス、`where` 節、関数の仮引数部分は、準クォートの実現のために内部的に使用されます。
- マクロの変数置換 `$x` を表すためのトークン。内部的には `$` を含まない形で保存さ

れます。構文文脈も保存されます。

4.2 構文解析

Rust の構文解析器は、大まかにいうと次の 2 つの仕事を行います。

- トークン列を抽象構文木に変換する。
- `mod foo;` がある場合は、別のファイルを新規に読み出して再帰的に字句解析と構文解析を呼び出す。

後者は、C 言語でいうところの `#include` と似ていると考えてよいでしょう。ただし、Rust ではプリプロセッサではなく構文解析器がこれを行います。

4.2.1 構文解析のための字句

Rust のトークン列は、構文解析器から見ると大雑把に以下のように分類されます。

ここでは簡単のために、記号やキーワードなどのトークンは単に `"*="` や `"fn"` といった文字列として表記します。実際には単一のトークンにマッチすると考えてください。

```

1 Token ::= NonParen | "(" | ")" | "{" | "}" | "[" | "]"
2
3 NonParen ::=
4     InnerDocComment
5     | OuterDocComment
6     | "=" | "<" | "<=" | "==" | "!=" | ">=" | ">"
7     | "&&" | "||" | "!" | "~"
8     | "+" | "-" | "*" | "/" | "^" | "&" | "|" | ">>" | "<<"
9     | "+=" | "-=" | "*=" | "/=" | "^="
10    | "&=" | "|=" | ">>=" | "<<="
11    | "@" | "." | ".." | "..." | "," | ";" | ":"
12    | "::" | "->" | "<-" | "=>" | "#" | "$" | "?"
13    | "_"
14    | IntegerLiteral | FloatLiteral
15    | ByteLiteral | ByteStringLiteral
16    | CharLiteral | CharStringLiteral
17    | Identifier
18    | Lifetime
19    | "as" | "box" | "break" | "const" | "continue" | "crate"
20    | "else" | "enum" | "extern" | "false" | "fn" | "for"
21    | "if" | "impl" | "in" | "let" | "loop" | "match" | "mod"
22    | "move" | "mut" | "pub" | "ref" | "return" | "self" | "Self"
23    | "static" | "struct" | "super" | "trait" | "true" | "type"
24    | "unsafe" | "use" | "where" | "while"
25    | "" | "{{root}}" | "$crate"
26    | NtItem | NtBlock | NtStmt | NtPat | NtExpr | NtTy | NtIdent
27    | NtMeta | NtPath | NtTT
28    | NtArm | NtImplItem | NtTraitItem | NtGenerics

```

```

29 | NtWhereClause | NtArg
30 | SubstNt

```

- 記号や強キーワードはそれぞれ別種のトークンとして扱われます。
- 弱キーワード `default`, `union`, `'static` は原則として識別子や生存期間の一種として扱われますが、特定の文脈では特別扱いされます。
- リテラルの種類は構文解析にとってはそれほど重要ではありませんが、整数リテラルと文字列リテラルは特別な扱いを受けることがあります。

4.2.2 列の文法

Rust の構文では、コンマ区切りの列が様々な場所出てきます。これらの多くは、0 個以上の要素が指定可能で、末尾に最大 1 個の余分なコンマをつけることが許されています。

この節では、このパターンを、以下のように定義される `sequence(E, D)` を用いて略記します。

```

1 sequence(E, D) ::= (E (D E)* D)?

```

4.2.3 crate

```

1 Crate ::= InnerAttribute* Item*

```

- `crate` 全体は `mod` とほぼ同じですが、構造上外部属性が指定できません。

4.2.4 アイテム

```

1 Item ::= OuterAttribute* (NtItem | Visibility ItemBody)
2
3 ItemBody ::=
4     ItemExternCrate
5     | ItemUse
6     | ItemStatic
7     | ItemConst
8     | ItemFn
9     | ItemMod
10    | ItemForeign
11    | ItemType
12    | ItemEnum
13    | ItemStruct
14    | ItemUnion
15    | ItemTrait
16    | ItemImpl
17    | ItemMacro
18
19 ItemMacro ::= ItemLikeMacroInvoc

```

- 関数や型など、**Rust** でグローバルな位置に置けるものはアイテムと呼ばれます。
- アイテムには属性と可視性を付与することができます。内部属性を書けるアイテムについては、内部属性と外部属性はマージされます。
- **Visibility** に **?** がついていませんが、これは可視性を省略できないという意味ではありません。**Visibility** 自身の定義に、省略可能であることが含まれています。
- 構文解析済みのトークン (**NtItem**) が来た場合はそれをそのまま受理します。**NtItem** には可視性を付与することはできません。**NtItem** に由来する属性と、**NtItem** に付与した外部属性は、マージされます。
- **unsafe** や **extern** などのキーワードを前置できるアイテムがあるため、アイテムの種類を決定するには数トークンほど先読みをする必要があります。
- アイテム (トレイト内アイテム、実装内アイテム、外部アイテムを含む) の構文上の特徴として、波括弧もしくはセミコロンで終端されるというものがあります。なお、セミコロン単体はアイテムではありません。
- アイテム (トレイト内アイテム、実装内アイテム、外部アイテムを含む) のもう一つの構文上の特徴として、マクロ以外はキーワード (弱キーワードを含む) から開始されるというものがあります。

4.2.5 外部 crate 宣言

```
1 ItemExternCrate ::=
2 "extern" "crate" Identifier ("as" Identifier)? ";"
```

- **crate** のルートモジュールを **as** で指定した名前参照します。
- **as** を省略した場合は、**crate** 名が使われます。

4.2.6 インポート

```
1 ItemUse ::=
2 "use" ModPath ("as" Identifier)? ";"
3 | "use" ModPath? "::" "*" ";"
4 | "use" (ModPath? "::")? "{" sequence(UseListElement, ",") "}" ";"
5
6 UseListElement ::= (Identifier | "self") ("as" Identifier)?
```

- 定義に対する別名づけを行います。単独インポートとグロブインポートの2種類があります。
- **{ }** で囲んだものはリストインポートといい、複数の単独インポートに展開されます。**self** を使うことで、モジュールそのものとモジュール内の定義を同時にインポートすることもできます。
- 単独インポートで **as** を省略した場合は、パスの最後の要素が採用されます。
- パスが通常の識別子から始まる場合は、強制的に絶対パスとして解釈されます。

4.2.7 静的アイテムと定数アイテム

```

1 ItemStatic ::= "static" "mut"? Identifier ":" Type "=" Expression ";"
2
3 ItemConst  ::= "const" Identifier ":" Type "=" Expression ";"

```

- 定数または静的記憶期間を持つ領域を定義します。定数に `mut` はつけられません。
- `let` とは異なり、型は必須です。また、`let` のように左辺にパターンは使えません。

4.2.8 関数定義

```

1 ItemFn ::=
2   FnPrefix "fn" Identifier Generics?
3   "(" sequence(Pattern ":" Type, ",") ")" ("->" Type)?
4   WhereClauses?
5   Block
6
7 FnPrefix ::=
8   "unsafe"? ("extern" CharStringLiteral)?
9   | "const" "unsafe"?

```

- 関数を定義します。
- `feature(const_fn)/RFC0911 [3]` により `const` が指定できます。指定すると、定数文脈でその関数を利用できるようになります。
- `unsafe` を指定すると、この関数の利用は `unsafe` だとマークされます。同時に、関数内部では `unsafe` 機能の利用が可能になります。
- `extern` により ABI を指定できます。 `extern` を省略すると `"rust"` ABI になります。 `extern` とだけ指定すると `"C"` ABI になります。
- `const` と `extern` は同時指定できません。
- ジェネリックスおよび `where` が指定できます。ジェネリックスの位置は関数名の直後で、`where` の位置は戻り値型の直後です。
- 引数リストはパターンと型を並べたものです。他の `fn` に比べて最も単純といえます。
- ブロック内に内部属性があれば、それは関数の属性になります。

4.2.9 モジュール

```

1 ItemMod ::=
2   "mod" Identifier ";" /* Parser treats it specially. */
3   | "mod" Identifier "{" InnerAttribute* Item* "}"
4
5 ModuleFile ::= InnerAttribute* Item*

```

- `mod foo {}` はモジュールを定義します。中には内部属性とアイテムを書くことができます。

- `mod foo;` のようにセミコロン終端された `mod` は、それ自身はモジュールの中身を持っていません。そこで、構文解析器はその場で別のファイルを開いて、そこからモジュールの中身を読み取ります。
- `mod foo;` の処理は、このアイテムに付属している外部属性に影響を受けます。ファイルを発見するまでは内部属性は入手できませんから、ファイル発見にかかわる設定は外部属性として書かなければいけません。
- 通常、属性の内容は構文解析より後に処理されますが、`mod` に付与された `cfg_attr(...)` はその場で処理されます。(`mod foo {}` でも処理されますが、これが仕様かどうかはわかりません。)
- さらに、`mod foo;` 形式の場合は、`cfg(...)` がその場で処理されます。偽と評価された場合はファイルの探索を行いません。(`mod foo {}` では処理されませんが、これが仕様かどうかはわかりません。)
- `mod foo;` のファイル発見規則はやや複雑です。まず、ローカル `crate` の各モジュールにはディレクトリが紐付けられています。また、そのモジュールがディレクトリを「所有しているかどうか」というフラグも管理されています。これに基いて以下のようにファイルが発見されます。
 - `crate` のルートモジュールのファイルとディレクトリは、コンパイラドライバによって指定される。例えば `src/main.rs` と `src/` 所有 が指定される。
 - `mod foo {}` は親モジュールのディレクトリの直下の `foo` が紐付けられる。所有関係は親モジュールから継承される。
 - ただし、`#[path = "somewhere"]` (ディレクトリ) が指定された場合は、親モジュールからの相対パスでこのディレクトリが紐付けられ、このディレクトリを所有していることになる。
 - ブロックには、それを含むモジュールと同じディレクトリが紐付けられるが、そのディレクトリを所有はしない。
 - `mod foo;` はまずファイルを探る。親モジュールのディレクトリの直下の `foo.rs` と `foo/mod.rs` のうち、存在しているほうを採用する。どちらもなかったり、どちらも存在している場合はエラーとなる。また、親モジュールがディレクトリを所有していない場合はエラーとなる。
 - ただし、`#[path = "somewhere"]` (ファイル) が指定された場合は、上記のファイルを探るせずに、指定されたファイルを採用する。親モジュールがディレクトリを所有していなくてもエラーとはならない。
 - 採用されたファイルの直上のディレクトリが紐付けられる。ファイル名が `mod.rs` だった場合のみ、所有しているものとみなされる。
 - マクロ呼び出しの実引数がパースされるさいは、マクロ呼び出しの位置でパースされたかのように振る舞う。マクロ展開結果がパースされるさいも、マクロ呼び出しの位置でパースされたかのように振る舞う。

4.2.10 外部アイテムの参照

```

1 ItemForeign ::=
2 "extern" CharStringLiteral?
```

```

3  "{" InnerAttribute* ForeignItem* "}"
4
5  ForeignItem ::=
6      OuterAttribute* Visibility
7      (ForeignItemStatic | ForeignItemFn)
8
9  ForeignItemStatic ::=
10     "static" "mut"? Identifier ":" Type ";"
11
12 ForeignItemFn ::=
13     "fn" Identifier Generics?
14     ForeignItemFnArgs ("->" Type)?
15     WhereClauses? ";"
16
17 ForeignItemFnArgs ::=
18     "(" sequence(Pattern ":" Type, ",") ")"
19     | "(" (Pattern ":" Type ",")+ "... " ")"

```

- 外部で定義されたシンボルをアイテムとして参照するには **extern** { } で囲みます。ABI 名を省略した場合は "C" として扱われます。
- 静的変数や関数の定義と似ていますが、定義部分がありません。
- **fn** に **const**, **unsafe**, **extern** などを追加で指定することはできません。
- 個別のアイテムに可視性をつけることができます。なお、**extern** 自身に可視性をつけても構文解析後にエラーになります。
- 関数定義と同様、引数名は必要です。引数はパターンとしてパースされますが、識別子 (**mut** なし) とワイルドカード以外のパターンを指定すると構文解析後にエラーになります。
- 引数リストの末尾に ... を指定することで可変長引数を表現できます。可変長部分の前に 1 つ以上の引数が必要です。

4.2.11 型別名

```

1  ItemType ::=
2      "type" Identifier Generics? WhereClauses? "=" Type ";"

```

- 型別名を定義します。型別名は、型検査では元の型とほぼ同じものとして扱われます。

4.2.12 列挙型

```

1  ItemEnum ::=
2      "enum" Identifier Generics? WhereClauses?
3      "{" (EnumBody | EnumBodyWithDiscriminator) "}"
4
5  EnumBody ::=

```

```

6  sequence(
7      RecordEnumVariant | TupleEnumVariant | UnitEnumVariant,
8      ",")
9
10 EnumBodyWithDiscriminator ::=
11     sequence(
12         UnitEnumVariant | EnumBodyWithDiscriminator,
13         ",")
14
15 RecordEnumVariant ::=
16     OuterAttribute* Identifier "{" RecordStructBody "}"
17
18 TupleEnumVariant ::=
19     OuterAttribute* Identifier "(" TupleStructBody ")"
20
21 UnitEnumVariant ::=
22     OuterAttribute* Identifier
23
24 EnumVariantWithDiscriminator ::=
25     OuterAttribute* Identifier "=" Expression

```

- 代数的データ型を定義します。代数的データ型の各腕はバリエーションと呼ばれます。
- バリエーションの定義は構造体とほぼ同様です。ユニット形式、タプル形式、レコード形式があります。
- ユニット形式のバリエーションのみからなる列挙型は C 風列挙型 (C-like enum) と呼ばれます。C 風列挙型は C 言語の列挙型とほぼ同様です。
- 特別なバリエーションとして、判別子 (discriminator) を明示するバリエーションがあります。判別子の明示は C 風列挙型でのみ使えます。この判定は構文解析の途中に行われますが、これが仕様かどうかはよくわかりません。

4.2.13 構造体

```

1  ItemStruct ::=
2      "struct" Identifier Generics?
3      ( WhereClauses? ";"
4      | WhereClauses? "{" RecordStructBody "}"
5      | "(" TupleStructBody ")" WhereClauses? ";" )
6
7  RecordStructBody ::= sequence(StructDeclField, ",")
8
9  StructDeclField ::=
10     OuterAttribute* Visibility Identifier ":" Type
11
12  TupleStructBody ::=
13     sequence(OuterAttribute* Visibility Type, ",")

```

- 構造体型を定義します。C 言語と同様の、複数のフィールドを横に並べてできる型です。
- 構造体はその形式により 3 種類に分けられます。ユニット構造体、タプル構造体、レコード構造体です。
- ユニット構造体はフィールドリストを持ちません。したがってフィールドは必ず 0 個です。
- タプル構造体は丸括弧で囲まれたフィールドリストを持ちます。各フィールドは名前を持たず、0,1,2,... という番号で識別されます。
- レコード構造体は波括弧で囲まれたフィールドリストを持ちます。各フィールドは名前を持ちます。

4.2.14 共用体

```

1 ItemUnion ::=
2   "union" Identifier Generics? WhereClauses?
3   "{" RecordStructBody "}"

```

- `feature(untagged_unions)/RFC1444 [4]` が提供する `unstable` 機能です。
- 共用体型を定義します。C 言語と同様の、複数のフィールドを同じ先頭アドレスに割り当ててできる型です。
- 構文上は、共用体はレコード構造体とほぼ同じです。
- `union` は弱キーワードであるため、これ自身は識別子と紛らわしいです。しかし、Rust の構文ではキーワードでない識別子が 2 つ並ぶことはないため、曖昧性なく `union` を発見することができます。

4.2.15 トレイト

```

1 ItemTrait ::=
2   "unsafe"? "trait" Identifier Generics?
3   (":" TyParamBounds)? WhereClauses?
4   "{" TraitItem* "}"
5
6 TraitItem ::= OuterAttribute* TraitItemBody
7
8 TraitItemBody ::=
9   TraitItemConst
10  | TraitItemMethod
11  | TraitItemType
12  | TraitItemMacro
13
14 TraitItemConst ::=
15   "const" Identifier ":" Type ("=" Expression)? ";"
16
17 TraitItemMethod ::=

```

```

18 FnPrefix "fn" Identifier Generics? WhereClauses?
19 TraitMethodArgs ("->" Type)?
20 (";" | Block)
21
22 TraitMethodArgs ::=
23     "(" sequence(Pattern ":" Type, ",") ")"
24 | "(" SelfArgument ")"
25 | "(" SelfArgument ", " sequence(Pattern ":" Type, ",") ")"
26
27 SelfArgument ::=
28     split_here("&") Lifetime? "mut"? "self"
29 | "mut"? "self" (":" Type)?
30
31 TraitItemType ::= "type" TyParam ";";
32
33 TraitItemMacro ::= ItemLikeMacroInvoc

```

- トレイトを定義します。 `unsafe` を指定することができます。
- ジェネリックスと `where` のほかに、スーパートレイト境界を指定できます。これは `Self` に対する境界として解釈されます。
- トレイト内アイテムは、マクロの他には、関連定数、メソッド、関連型の 3 種類が指定できます。
- 関連定数は `feature(associated_consts)/RFC0195 [2]` が提供する `unstable` 機能です。
- 関連型は単相です。
- メソッドには実装を与えることもできます。実装のないメソッドを必須メソッド、実装のあるメソッドを提供メソッドといいます。
- メソッドは第一引数に `self` という特殊な名前の引数をとることが可能です。型を省略するための短縮記法もあります。
- 現在の実装では、(メソッドの実装が提供されているか否かによらず) メソッドの引数名は省略できます。その影響で、メソッドの引数部を書けるパターンに構文上の制約が存在します。この構文は RFC1685 [5] で廃止が決定されました。

4.2.16 実装

```
1 ItemImpl ::= ...
```

4.2.17 型

```
1 Type ::= ...
```

4.2.18 式

```
1 Expression ::= ...
```

4.2.19 パターン

```
1 Pattern ::= ...
```

4.2.20 ブロック

```
1 Block ::= ...
```

4.2.21 文

```
1 Statement ::= ...
```

4.2.22 ジェネリックス

```
1 Generics ::= ...
```

```
2
```

```
3 TyParam ::= ...
```

```
4
```

```
5 WhereClauses ::= ...
```

```
6
```

```
7 TyParamBounds ::= ...
```

4.2.23 マクロ呼び出し

```
1 ItemLikeMacroInvoc ::= ...
```

```
2
```

```
3 StmtLikeMacroInvoc ::= ...
```

```
4
```

```
5 ExprLikeMacroInvoc ::= ...
```

4.2.24 パス

```
1 ModPath ::= ...
```

4.2.25 可視性

```
1 Visibility ::= ...
```

4.2.26 属性

```
1 OuterAttribute ::= ...
```

```
2
```

```
3 InnerAttribute ::= ...
```

第 5 章

構文拡張

5.1 構文拡張のデータベース

5.2 マクロ

5.3 組み込みマクロ

5.4 `macro_rules`

5.5 属性構文拡張

5.6 手続きマクロ

第 6 章

名前解決

6.1 モジュール

6.2 インポート解決

6.3 レキシカルスコープの解決

6.4 型依存の名前解決

6.5 生存期間の解決

第 7 章

型検査

7.1 型

7.2 トレイト

7.3 Hindley-Milner 型推論

7.4 トレイト選択

7.5 リージョン推論

7.6 ドロップ検査

7.7 借用検査

7.8 可変性検査

第 8 章

コード生成

8.1 MIR

8.2 LLVM IR

第 9 章

ドキュメンテーション

9.1 doc-comment の構文

参考文献

- [1] Rust issue #28979: tracking issue for non-ascii identifiers. <https://github.com/rust-lang/rust/issues/28979>.
- [2] Rust RFC #0195: associated items. <https://github.com/rust-lang/rfcs/blob/master/text/0195-associated-items.md>.
- [3] Rust RFC #0911: const fn. <https://github.com/rust-lang/rfcs/blob/master/text/0911-const-fn.md>.
- [4] Rust RFC #1444: union. <https://github.com/rust-lang/rfcs/blob/master/text/1444-union.md>.
- [5] Rust RFC #1685: deprecate anonymous parameters. <https://github.com/rust-lang/rfcs/blob/master/text/1685-deprecate-anonymous-parameters.md>.
- [6] Mark Davis. UAX #31: Unicode identifier and pattern syntax. Technical report, Unicode Inc., 2016. <http://www.unicode.org/reports/tr31/tr31-25.html>.
- [7] Mark Davis, Laurențiu Iancu, and Ken Whistler. UAX #44: Unicode character database. Technical report, Unicode Inc., 2016. <http://www.unicode.org/reports/tr44/tr44-18.html>.