Rust コンパイラーウォークスルー

原 将己

2017年5月4日

目次

第 1 章	はじめに	5
第2章	Rust コンパイラーの開発フロー	7
2.1	RFC	7
2.2	メインリポジトリ	7
2.3	文書	7
2.4	フォーラム	7
第3章	Rustc と Crate と Cargo	9
3.1	crate のソース	9
3.2	crate のバイナリー	9
3.3	Cargo の基本動作	9
3.4	依存関係の解決	9
3.5	crates リポジトリ	9
第4章	字句と構文	11
4.1	字句解析	11
4.2	構文解析	21
第 5 章	構文拡張	25
5.1	構文拡張のデータベース	25
5.2	マクロ	25
5.3	組み込みマクロ	25
5.4	macro_rules	25
5.5	属性構文拡張	25
5.6	手続きマクロ	25
第6章	名前解決	27
6.1	モジュール	27
6.2	インポート解決	27
6.3	レキシカルスコープの解決	27
6.4	型依存の名前解決	27
6.5	生存期間の解決	27
第7章	型検査	29

目次

7.1	型	29
7.2	トレイト	29
7.3	Hindley-Milner 型推論	29
7.4	トレイト選択	29
7.5	リージョン推論	29
7.6	ドロップ検査	29
7.7	借用検査	29
7.8	可変性検査	29
第8章	コード生成	31
8.1	MIR	31
8.2	LLVM IR	31
第 9 章	ドキュメンテーション	33
9.1	doc-comment の構文	33
参考文献		35

第 1 章

はじめに

第2章

Rust コンパイラーの開発フロー

- 2.1 RFC
- 2.2 メインリポジトリ
- 2.3 文書
- 2.4 フォーラム

第3章

Rustc と Crate と Cargo

- 3.1 crate のソース
- 3.2 crate のバイナリー
- 3.3 Cargo の基本動作
- 3.4 依存関係の解決
- 3.5 crates リポジトリ

第4章

字句と構文

4.1 字句解析

Rust のソースファイルはまず字句解析にかけられます。

4.1.1 凡例

この節では BNF 風の記法を用いて文法を記述しています。BNF 風ですが実際には PEG (解析表現文法) に近い意味論です。

- A B は連接です。 2 つの構文要素をこの順に並べたものです。
- A | Bは PEG の順序つき OR (A / B) です。 A を試し、失敗したら B を試します。 ただし、この節で A | B と書いた場合は、順番による違いが生じないようになっているはずです。
- A*, A+, A? は、0回以上の繰り返し、1回以上の繰り返し、高々1回の出現です。いずれも貪欲に実行したものと解釈しますが、それを知らなくても読めるように配慮したつもりです。
- lookahead A と not A は先読み肯定と先読み否定です。慣れていないと読み間違えやすいかもしれません。 A (not B) は、 A だが B が後続しないものです。 (not B) A は、 A だが同じ場所で B にはならないものです。
- eof は文字列の末尾です。
- "+=" のようにダブルクオートで囲んだものは文字列で、1 文字ずつ連接したもの と同じです。C 言語風のエスケープを使っています。

4.1.2 Unicode

Rust の字句解析器は UTF-8 でエンコードされたファイルを入力します。 UTF-8 をデコードしたものは、Unicode スカラー値の列とみなすことができます。

Unicode スカラー値とは、 0 以上 0xD800 未満か 0xE000 以上 0x110000 未満の整数のことです。ここでは、Unicode 文字といったら Unicode スカラー値のことを指すことにします。

AnyChar ::= /* Any Unicode scalar value */

```
3 AnyAsciiChar ::=
4  /*Any Unicode scalar value < 128 */</pre>
```

4.1.3 トークン列

Rust の字句解析器の仕事は、Unicode スカラー値の列を受け取り、字句エラーがなければトークン列を返すことです。

```
1 File ::= MaybeShebang Skip (Token Skip)* eof
2
3
   Token ::=
4
       InnerDocComment
5
     | OuterDocComment
    | Symbol
6
7
    | Literal
8
    | Identifier
9
     | StrictKeyword
10
     | ReservedKeyword
11
     | Lifetime
```

• 字句的な特徴からトークンを分類すると、doc-comment, 記号、数値リテラル、文字列系リテラル、識別子とキーワード、生存期間と分けられます。しかし、構文解析器やマクロ展開器にとっての分類は必ずしもこれに沿った形ではありません。

4.1.4 空白とコメント

Rust の字句解析器は空白とコメントを無視します。コンパイラにとっては、これらはトークンを明示的に分割する以上の役割はありません。ただし、doc-comment は特別な扱いを受けます。

```
1 Skip ::= (Whitespace | Comment)* (not MaySkip)
2
3
   Whitespace ::= PatternWhitespace+ (not PatternWhitespace)
4
5
   PatternWhitespace ::=
6
    /* Any Unicode scalar value with PATTERN_WHITE_SPACE */
7
   Comment ::= "//" (not ("/" | "!")) LineCommentChar* ("\n" | eof)
8
9
         | "///" LineCommentChar* ("\n" | eof)
             | "/*" (not ("*" | "!")) BlockCommentBody* "*/"
10
          | "/**/"
11
12
   LineCommentChar ::= (not "\n") AnyChar
   BlockCommentBody ::= (not ("/*" | "*/")) AnyChar
14
                      | "/*" BlockCommentBody* "*/"
15
16
17
   MaySkip ::= PatternWhitespace
```

4.1 字句解析 13

UAX #31 [1] に従い、Pattern_White_Space に分類される文字は全て空白文字とみなされます。Pattern_White_Space は UAX #44 [2] で規定されており、PropList.txtで取得できます。現時点では以下の文字がPattern_White_Spaceに分類されます。

- U+0009 Control: CHARACTER TABULATION

- U+000A Control: LINE FEED (LF)

- U+000B Control: LINE TABULATION

- U+000C Control: FORM FEED (FF)

- U+000D Control: CARRIAGE RETURN (CR)

- U+0020 SPACE

- U+0085 Control: NEXT LINE (NEL)

- U+200E LEFT-TO-RIGHT MARK

- U+200F RIGHT-TO-LEFT MARK

- U+2028 LINE SEPARATOR

- U+2029 PARAGRAPH SEPARATOR

- C と同様に // と /* がコメントとみなされますが、 doc-comment はトークンとみ なされるため、この意味で通常のコメントとは異なります。
- Cとは異なり、/* はネストさせることができます。
- //// 型のコメントの中では、doc-comment と同様に、単独でのキャリッジリターンの出現が禁止されています。これはバグではないかと思います。
- 1 行目が #! で始まる場合、その行はシバンとみなされ、コメントと同様に無視されます。ただし、 #![で始まる場合は、内部属性と紛らわしいため、例外的にシバンとはみなされません。

4.1.5 doc-comment

doc-comment はコメントとよく似た構文を持ちますが、Rust コンパイラはこれを一つのトークンとみなします。

```
InnerDocComment ::=

"//!" LineDocCommentChar* ("\r\n" | "\n" | eof)

"/*!" BlockDocCommentBody* "*/"

OuterDocComment ::=

"///" (not "/") LineDocCommentChar* ("\r\n" | "\n" | eof)
```

- ///, //!, /**, /*! で始まるコメントは、doc-comment と解釈されます。
- 例外として、 //// で始まるコメントと /**/ は、doc-comment ではありません。
- これらの doc-comment は、外部ツールだけではなく、Rust コンパイラによって認識されます。字句的に正しい doc-comment であっても、構文的に誤った位置にあれば、コンパイルエラーになる可能性があります。
- 通常のコメントと異なり、doc-comment 内ではキャリッジリターン (CR) が単独で 出現してはいけません。必ずラインフィードと対で (CRLF) 出現する必要があり ます。
- Cとは異なり、/* はネストさせることができます。

4.1.6 記号

```
Symbol ::=
        "=" (not ("=" | ">"))
 2
3
      | "<" (not ("=" | "<" | "-"))
      | "<=" | "==" | "!=" | ">="
 4
 5
      | ">" (not ("=" | ">"))
      | "&&" | "||"
 6
 7
      | "!" (not "=")
      | "~"
8
 9
      | "+" (not "=")
10
      | "-" (not ("=" | ">"))
11
      | "*" (not "=")
12
      | "/" (not "=")
13
      | "^" (not "=")
      | "&" (not ("=" | "&"))
14
15
      | "|" (not ("=" | "|"))
16
      | ">>" (not "=")
      | "<<" (not "=")
17
      | "+=" | "-=" | "*=" | "/=" | "^="
18
19
      | "&=" | "|=" | ">>=" | "<<="
20
      | "@"
21
      | "." (not ".")
      | ".." (not ".")
22
23
      | "..." | "," | ";"
24
      | ":" (not ":")
      | "::" | "->" | "<-" | "=>" | "#"
25
```

4.1 字句解析 **15**

```
26 | "$" | "?"

27 | "(" | ")" | "{" | "}" | "[" | "]"

28 | Underscore
```

- 記号のパース規則は単純です。記号の候補の中で、一番長いものを選択します。
- アンダースコアだけ特殊です。アンダースコアは字句的には識別子に近いですが、 パーサーに渡す段階では識別子よりも記号に近い扱いを受けます。

4.1.7 リテラル

```
Literal ::= LiteralBody LiteralSuffix?

LiteralBody ::= NumberLiteralBody | StringLikeLiteralBody

LiteralSuffix ::= IdentifierOrKeyword | Underscore
```

- リテラルは、数値リテラルと、文字列系リテラルの2種類あります。
- いずれのリテラルにも、識別子またはキーワードを後置することができます。これは型を表すのに使います。
- 識別子またはキーワードのかわりにアンダースコアを後置することもできます。リテラルの構造上数値リテラルでは不可能です。後置しても、パーサーからはなかったものとして扱われます。もしかしたらバグかもしれません。
- 後置された文字列は型として解釈されますが、これは構文解析器の仕事です。

4.1.8 数値リテラル

```
NumberLiteralBody ::= IntegerLiteralBody | FloatLiteralBody
1
2
3
   IntegerLiteralBody ::=
4
        "Ob" "_"* BinDigit BinDigitU* (not DecDigitU) EnsureNotFloat
     | "Oo" "_"* OctDigit OctDigitU* (not DecDigitU) EnsureNotFloat
5
     | Decimal (not DecDigitU) EnsureNotFloat
6
7
     | "0x" "_"* HexDigit HexDigitU* (not HexDigitU) EnsureNotFloat
8
   EnsureNotFloat ::= not ("." | "e" | "E")
9
10
                     | lookahead ".."
11
                     | lookahead ("." IdentStart)
12
13
   FloatLiteralBody ::=
        Decimal "." (not ("." | IdentStart | DecDigit))
14
15
     | Decimal "." Decimal (not ("e" | "E"))
      | Decimal "." Decimal ExponentPart
16
17
      | Decimal ExponentPart
18
19
   ExponentPart ::=
        ("e" | "E") ("-" | "+")?
20
```

```
"_"* DecDigit DecDigitU* (not DecDigitU)
21
22
23
   Decimal ::= DecDigit DecDigitU*
24
25
   BinDigit ::= "0" | "1"
26
27
   OctDigit ::= "0" | "1" | "2" | "3" | "4" | "5" | "6" | "7"
28
   DecDigit ::= OctDigit | "8" | "9"
29
30
31
   HexDigit ::=
       DecDigit | "A" | "B" | "C" | "D" | "E" | "F"
32
33
      | "a" | "b" | "c" | "d" | "e" | "f"
34
35
   BinDigitU ::= BinDigit | "_"
36
   OctDigitU ::= OctDigit | "_"
37
38
39
   DecDigitU ::= DecDigit | "_"
40
   HexDigitU ::= HexDigit | "_"
```

- 数値リテラルは、整数リテラルと浮動小数点数リテラルからなります。
- 負の符号は単なる単項演算子です。トークンとしては、非負の数のみ扱われます。
- アンダースコアで数を区切ることができます。しかし、数字がないのにアンダースコアだけある、という状況は禁止されています。また、リテラルの先頭や、小数点の直後には、アンダースコアを置くことはできません。
- 2 進数、8 進数、16 進数では接頭辞を明示する必要があります。 **0** で始まっていて も 10 進数です。また、接頭辞は小文字でなければなりません。16 進数の桁や、指数部をあらわす **e** は大文字でも小文字でも構いません。
- 小数点の手前を省略することはできません。また、小数点の直後を省略できるのは、直後が. でも識別子でもない場合だけです。
- 整数の直後に.を置けるのは、その直後が.または識別子の場合だけです。
- 上記の規則について、現在は._ が特別扱いされています。おそらくバグではないかと思います。
- また、現在は **0e**, **0E** から始まる浮動小数点数がパースエラーになるバグがあります。これは修正済みで、次の安定板には入ると思います。
- a から f までの文字は、16 進数以外では、トークン区切りとみなされます。しかし、たとえ8進数であっても、9 はトークン区切りとはみなされません。この位置でパースエラーになります。
- f という文字は、微妙なバランスの上に成立しています。16 進数は整数でしか使えないので、接尾辞にはi かu が指定できれば十分です。いっぽう、f を含む接尾辞が指定される浮動小数点数では、f の部分でリテラル本体のパースが打ち切られるため、うまくf を指定することができます。

4.1 字句解析 **17**

4.1.9 文字列系リテラル

```
1
   StringLikeLiteralBody ::=
2
       ByteLiteralBody | ByteStringLiteralBody
3
     | CharLiteralBody | CharStringLiteralBody
4
5
   ByteLiteralBody ::=
       "b\'" (not ("\\" | "\t" | "\r" | "\'")) AnyAsciiChar "\'"
6
7
      | "b\'\\" ("n" | "r" | "t" | "\\" | "\'" | "\"" | "0") "\'"
8
     | "b\'\x" HexDigit HexDigit "\'"
9
10
   ByteStringLiteralBody ::=
11
       "b\"" SingleByte* "\""
12
      | "br" repeated("#", N) "\""
         ((not ("\"" repeated("#", N))) AnyAsciiChar)*
13
14
         "\"" repeated("#", N)
15
         /* N >= 0 */
16
         /* repeated("#", 3) means e.g. "###" */
17
18
   SingleByte ::=
       (not ("\\" | "\r" | "\"")) AnyAsciiChar
19
20
     | "\r\n"
21
     | "\\" ("n" | "r" | "t" | "\\" | "\'" | "\"" | "0")
22
     | "\\" ("\r\n" | "\n") Whitespace
     | "\\x" HexDigit HexDigit
23
24
25
   CharLiteralBody ::=
26
       "\'" (not ("\\" | "\t" | "\n" | "\r" | "\'")) AnyChar "\'"
27
     | "\'\\" ("n" | "r" | "t" | "\\" | "\'" | "\"" | "0") "\'"
28
     | "\'\x" OctDigit HexDigit "\'"
29
     | "\'\u{" HexDigit{1, 6} "}\'"
30
       /* only when x < 0xD800 or 0xE000 < x < 0x110000 */
31
32
   CharStringLiteralBody ::=
33
       "\"" SingleChar* "\""
34
      | "r" repeated("#", N) "\""
35
         ((not ("\"" repeated("#", N))) AnyChar)*
36
         "\"" repeated("#", N)
37
         /* N >= 0 */
38
         /* repeated("#", 3) means e.g. "###" */
39
40
   SingleChar ::=
41
       (not ("\\" | "\r" | "\"")) AnyChar
42
43
     | "\\" ("n" | "r" | "t" | "\\" | "\\" | "\"" | "0")
44
      | "\\" ("\r\n" | "\n") Whitespace
```

- バイト、バイト列、文字、文字列のいずれかを表します。
- バイトまたは文字は、で囲み、バイト列または文字列は"で囲みます。
- バイトまたはバイト列をあらわすときは b を前置します。
- バイト列または文字列の場合は生リテラルを使うこともできます。生リテラルは"のかわりに r###"のような形で開始します。この場合最初に "### が出現した位置で終了になります。 # の個数は 0 個以上で、左右で揃える必要があります。
- バックスラッシュ、水平タブ、ラインフィード、キャリッジリターン、デリミタ("または'のうちリテラルを囲うのに使われているほう)以外の ASCII 文字は、そのまま書くことができます。
- ASCII 以外の Unicode 文字 (Unicode スカラー値) は、文字または文字列リテラル にそのまま書くことができます。
- タブ、ラインフィード、CRLF(キャリッジリターン+ラインフィード)は、生リテラルでないバイト列または文字列に直接書くことができます。CRLFはLFとみなされます。
- 生リテラル以外のリテラルでは、\n,\r,\t,\\,\',\",\0 がエスケープとして利用できます。
- 生リテラル以外のリテラルでは、\x に続けて16進法で2桁(アンダースコア不可) の整数を指定することで、1 バイトエスケープになります。ただし、128 より大き いバイトは、文字または文字列リテラルでは指定できません。
- 生リテラル以外の文字または文字列リテラルでは、 \u{} の中に 16 進法で 1 桁から 6 桁まで (アンダースコア不可) の整数を指定することで、Unicode エスケープになります。 ただし指定できるのは Unicode スカラー値、つまり 0 以上 0xD800 未満か、 0xE000 以上 0x110000 未満の整数のみです。
- 生リテラル以外のバイト列または文字列リテラルでは、バックスラッシュの直後に 改行 (LF または CRLF) を続けると、該当バックスラッシュから、連続する空白文 字が全てスキップされます。
- 生リテラルではエスケープは使用できず、全ての文字がそのまま解釈されます。 生バイト列リテラルでは、ASCII 文字以外 (128 以上のスカラー値) は使用できま せん。

4.1.10 識別子またはキーワード

```
Identifier ::=
(not (StrictKeyword | ReservedKeyword)) IdentifierOrKeyword

IdentifierOrKeyword ::=
(not (Underscore | MaybeString))
IdentStart IdentContinue* (not IdentContinue)
```

4.1 字句解析 19

```
8
   IdentStart ::=
9
     /* Any Unicode scalar value with XID_Start */ | "_"
10
11
   IdentContinue ::=
12
     /* Any Unicode scalar value with XID_Continue */ | "_"
13
   Underscore ::= "_" (not IdentContinue)
14
15
   MaybeString ::= "r\"" | "r#" | "b\"" | "b\'" | "br\"" | "br#"
16
17
   StrictKeyword ::= StrictKeywordString (not IdentContinue)
18
19
20
   StrictKeywordString ::=
       "as" | "box" | "break" | "const" | "continue" | "crate"
21
22
     | "else" | "enum" | "extern" | "false" | "fn" | "for"
     | "if" | "impl" | "in" | "let" | "loop" | "match" | "mod"
23
     | "move" | "mut" | "pub" | "ref" | "return" | "self" | "Self"
24
25
     | "static" | "struct" | "super" | "trait" | "true" | "type"
26
      | "unsafe" | "use" | "where" | "while"
27
28
   ReservedKeyword ::= ReservedKeywordString (not IdentContinue)
29
30
   ReservedKeywordString ::=
       "abstract" | "alignof" | "become" | "do" | "final"
31
32
      | "macro" | "offsetof" | "override" | "priv" | "proc"
33
     | "pure" | "sizeof" | "typeof" | "unsized" | "virtual"
34
      | "yield"
35
36
   WeakKeyword ::= WeakKeywordString (not IdentContinue)
37
   WeakKeywordString ::= "default" | "union"
38
```

- 識別子またはキーワードは、ASCII の数字、大文字、小文字、アンダースコアからなります。また最初の文字は ASCII の大文字、小文字、アンダースコアのいずれかでなければなりません。
- この規則は feature(non_ascii_idents) により Unicode に拡張されます。この場合、識別子は XID_Continue な文字またはアンダースコアからなります。また、最初の文字は XID_Start な文字またはアンダースコアでなければなりません。これは UAX #31 [1] を考慮していますが、現在の実装は完全とはいえません。 UAX #31 では NKFC 等の正規化が言及されていますが、現在のコンパイラは正規化を全くしません。どの正規化をするかなどを含めて、議論の途中の状態です。
- アンダースコア 1 文字からなる場合は、識別子ではなく記号として解釈されます。
- r", r#, b", b', br", br# のいずれかで始まる場合は、これらの r, b, br は識別子ではなく文字列系リテラルの開始として解釈されます。
- 「識別子またはキーワード」のうち、いくつかは「強キーワード (strict keyword)」 「予約キーワード (reserved keyword)」「弱キーワード (weak keyword)」のいずれか

に分類されています。強キーワードは特別な意味をもつもので、予約キーワードは 将来強キーワードとして使われる可能性を考慮して予約されているものです。

- 強キーワードでも予約キーワードでもないものを識別子といいます。弱キーワード は、原則として識別子として扱われますが、文脈によっては特別な意味を持ちます。
- 識別子またはキーワードは、マクロの観点からはキーワードか否かに関係なく同等 に扱われます。識別子またはキーワードと、生存期間には、構文文脈が付与されま す。アンダースコアは識別子またはキーワードではないため、構文文脈は付与され ません。
- 字句解析器は、空の構文文脈を付与します。

4.1.11 生存期間

```
1 Lifetime ::=
2    (not "\'" IdentStart "\'")
3    "\'" IdentifierOrKeyword
4
5 WeakLifetimeKeyword ::= "\'static" (not IdentContinue)
```

- ・ 、に識別子またはキーワードを後続させたものは、生存期間です。
 - ただし、文字リテラルとしても解釈できる場合は、文字リテラルが優先されます。
 - '_ は生存期間ではありません。
 - 'static は生存期間ですが、文脈によっては特別な意味を持ちます。
 - 識別子またはキーワードと、生存期間には、構文文脈が付与されます。
 - 字句解析器は、空の構文文脈を付与します。

4.1.12 特殊なトークン

字句解析器以外によってのみ生成される特殊なトークンがいくつかあります。

- 空文字列、 {{root}}, \$crate はいずれも、上で説明した「識別子またはキーワード」の条件を満たしていませんが、内部的に識別子として生成される場合があります。
- 同じく、空文字列が内部的に生存期間として生成される場合があります。なお、通 常の生存期間の名前は、内部的には、を含んだ形で保存されています。
- 補間トークンは、AST を1つのトークンとしてみなしたものです。
 - アイテム、ブロック、文、パターン、式、型、識別子を含む補間トークンは、 マクロの展開のために使われます。
 - メタアイテム、パス、トークンツリーを含む補間トークンは、属性構文拡張の 展開のために使われます。
 - パターンマッチの腕、実装内アイテム、トレイト内アイテム、ジェネリックス、where 節、関数の仮引数部分は、準クオートの実現のために内部的に使用されます。
- マクロの変数置換 \$x を表すためのトークン。内部的には \$ を含まない形で保存さ

4.2 構文解析 **21**

れます。構文文脈も保存されます。

4.2 構文解析

Rust の構文解析器は、大まかにいうと次の2つの仕事を行います。

- トークン列を抽象構文木に変換する。
- mod foo; がある場合は、別のファイルを新規に読み出して再帰的に字句解析と構 文解析を呼び出す。

後者は、C言語でいうところの #include と似ていると考えてよいでしょう。ただし、 Rust ではプリプロセッサーではなく構文解析器がこれを行います。

4.2.1 構文解析のための字句

Rust のトークン列は、構文解析器から見ると大雑把に以下のように分類されます。 ここでは簡単のために、記号やキーワードなどのトークンは単に "*=" や "fn" といった文字列として表記します。実際には単一のトークンにマッチすると考えてください。

```
Token ::= NonParen | "(" | ")" | "{" | "}" | "[" | "]"
1
2
3
   NonParen ::=
4
       InnerDocComment
5
     | OuterDocComment
     | "=" | "<" | "<=" | "!=" | ">=" | ">"
6
7
     | "&&" | "||" | "!" | "~"
8
     | "+" | "-" | "*" | "/" | "^" | "&" | "|" | ">>" | "<<"
9
     | "+=" | "-=" | "*=" | "/=" | "^="
10
     | "&=" | "|=" | ">>=" | "<<="
     | "@" | "." | "..." | "," | ";" | ":"
11
12
     | "::" | "->" | "<-" | "=>" | "#" | "$" | "?"
13
14
     | IntegerLiteral | FloatLiteral
15
     | StringLikeLiteral
16
     | Identifier
17
     | Lifetime
      | "as" | "box" | "break" | "const" | "continue" | "crate"
18
19
     | "else" | "enum" | "extern" | "false" | "fn" | "for"
      | "if" | "impl" | "in" | "let" | "loop" | "match" | "mod"
20
21
     | "move" | "mut" | "pub" | "ref" | "return" | "self" | "Self"
22
     | "static" | "struct" | "super" | "trait" | "true" | "type"
23
     | "unsafe" | "use" | "where" | "while"
     | "" | "{{root}}" | "$crate"
24
25
     | NtItem | NtBlock | NtStmt | NtPat | NtExpr | NtTy | NtIdent
26
      | NtMeta | NtPath | NtTT
27
      | NtArm | NtImplItem | NtTraitItem | NtGenerics
28
      | NtWhereClause | NtArg
```

29 | SubstNt

- 記号や強キーワードはそれぞれ別種のトークンとして扱われます。
- 弱キーワード default, union, 'static は原則として識別子や生存期間の一種として扱われますが、特定の文脈では特別扱いされます。
- リテラルの種類は構文解析にとってはそれほど重要ではありませんが、整数リテラルは特別な扱いを受けることがあります。

4.2.2 列の文法

Rust の構文では、コンマ区切りの列が様々な場所で出てきます。これらの多くは、0 個以上の要素が指定可能で、末尾に最大 1 個の余分なコンマをつけることが許されています。

この節では、このパターンを、以下のように定義される sequence(E, D) を用いて略記します。

```
1 sequence(E, D) ::= (E (D E)* D?)?
```

4.2.3 crate

- 1 Crate ::= InnerAttribute* Item*
 - crate 全体は mod とほぼ同じですが、構造上外部属性が指定できません。

4.2.4 アイテム

```
Item ::= OuterAttribute* (NtItem | Visibility ItemBody)
2
3
   ItemBody ::=
4
        ItemExternCrate
5
      | ItemUse
6
      | ItemStatic
7
      | ItemConst
8
     ItemFn
9
      | ItemMod
      | ItemForeign
10
     | ItemType
11
12
      | ItemEnum
13
      | ItemStruct
14
      | ItemUnion
15
      | ItemTrait
16
      | ItemImpl
17
      | ItemMacro
```

• 関数や型など、Rust でグローバルな位置に置けるものはアイテムと呼ばれます。

4.2 構文解析 23

• アイテムには属性と可視性を付与することができます。内部属性を書けるアイテムについては、内部属性と外部属性はマージされます。

- 構文解析済みのトークン (NtItem) が来た場合はそれをそのまま受理します。 NtItem には可視性を付与することはできません。
- unsafe や extern などのキーワードを前置できるアイテムがあるため、アイテム の種類を決定するには数トークンほど先読みをする必要があります。

4.2.5 外部 crate 宣言

```
1 ItemExternCrate ::=
2  "extern" "crate" Identifier ("as" Identifier)? ";"
```

- crate のルートモジュールを as で指定した名前で参照します。
- as を省略した場合は、crate 名が使われます。

4.2.6 インポート

```
1 ItemUse ::=
2    "use" ModPath ("as" Identifier)? ";"
3    | "use" ModPath? "::" "*" ";"
4    | "use" (ModPath? "::")? "{" sequence(UseListElement, ",") "}" ";"
5
6 UseListElement ::= (Identifier | "self") ("as" Identifier)?
```

- 定義に対する別名づけを行います。単独インポートとグロブインポートの 2 種類があります。
- { } で囲んだものはリストインポートといい、複数の単独インポートに展開されます。 self を使うことで、モジュールそのものとモジュール内の定義を同時にインポートすることもできます。
- 単独インポートで as を省略した場合は、パスの最後の要素が採用されます。
- パスが通常の識別子から始まる場合は、強制的に絶対パスとして解釈されます。

4.2.7 静的アイテム

```
1 ItemStatic ::= ...
```

4.2.8 定数アイテム

```
1 ItemConst ::= ...
```

4.2.9 関数宣言·関数定義

```
1 ItemFn ::= ...
```

4.2.10 モジュール

```
1 ItemMod ::= ...
  4.2.11 extern ブロック
1 ItemForeign ::= ...
  4.2.12 型別名
1 ItemType ::= ...
  4.2.13 列挙型
1 ItemEnum ::= ...
  4.2.14 構造体
1 ItemStruct ::= ...
  4.2.15 共用体
1 ItemUnion ::= ...
  4.2.16 トレイト
1 ItemTrait ::= ...
  4.2.17 実装
4.2.18 マクロ呼び出し
1 ItemMacro ::= ...
  4.2.19 パス
1 ModPath ::= ...
  4.2.20 可視性
1 Visibility ::= ...
  4.2.21 属性
1 OuterAttribute ::= ...
3 InnerAttribute ::= ...
```

第5章

構文拡張

- 5.1 構文拡張のデータベース
- 5.2 マクロ
- 5.3 組み込みマクロ
- 5.4 macro_rules
- 5.5 属性構文拡張
- 5.6 手続きマクロ

第6章

名前解決

- 6.1 モジュール
- 6.2 インポート解決
- 6.3 レキシカルスコープの解決
- 6.4 型依存の名前解決
- 6.5 生存期間の解決

第7章

型検査

- 7.1 型
- 7.2 トレイト
- 7.3 Hindley-Milner 型推論
- 7.4 トレイト選択
- 7.5 リージョン推論
- 7.6 ドロップ検査
- 7.7 借用検査
- 7.8 可変性検査

第8章

コード生成

- 8.1 MIR
- 8.2 LLVM IR

第9章

ドキュメンテーション

9.1 doc-comment の構文

参考文献

- [1] Mark Davis. UAX #31: Unicode identifier and pattern syntax. Technical report, Unicode Inc., 2016. http://www.unicode.org/reports/tr31/tr31-25.html.
- [2] Mark Davis, Laurențiu Iancu, and Ken Whistler. UAX #44: Unicode character database. Technical report, Unicode Inc., 2016. http://www.unicode.org/reports/tr44/tr44-18.html.